pragma solidity ^0.4.16;

/**
 * @title SafeMath
 * @dev Math operations with safety checks that throw on error
 */
library SafeMath {
  function mul(uint256 a, uint256 b) internal constant returns (uint256) {
    uint256 c = a * b;
    assert(a == 0 || c / a == b);
    return c;
  }

  function div(uint256 a, uint256 b) internal constant returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return c;
  }

  function sub(uint256 a, uint256 b) internal constant returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  function add(uint256 a, uint256 b) internal constant returns (uint256) {
    uint256 c = a + b;
    assert(c >= a);
    return c;
  }
}

/**
 * @title ERC20Basic
 * @dev Simpler version of ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/179
 */
contract ERC20Basic {
  uint256 public totalSupply;
  function balanceOf(address who) public constant returns (uint256);
  function transfer(address to, uint256 value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}

/**
 * @title ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/20
 */
contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender) public constant returns (uint256);
  function transferFrom(address from, address to, uint256 value) public returns (bool);
  function approve(address spender, uint256 value) public returns (bool);
  event Approval(address indexed owner, address indexed spender, uint256 value);
}

/**
 * @title Basic token
 * @dev Basic version of StandardToken, with no allowances.
 */
contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;

  /**
  * @dev transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[msg.sender]);

    // SafeMath.sub will throw if there is not enough balance.
    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public constant returns (uint256 balance) {
    return balances[_owner];
  }

}

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * @dev https://github.com/ethereum/EIPs/issues/20
 * @dev Based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) internal allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amount of tokens to be transferred
   */
  function transferFrom(address _from, address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[_from]);
    require(_value <= allowed[_from][msg.sender]);

    balances[_from] = balances[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    allowed[_from][msg.sender] = allowed[_from][msg.sender].sub(_value);
    Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   *
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(address _owner, address _spender) public constant returns (uint256 remaining) {
    return allowed[_owner][_spender];
  }

  /**
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   */
  function increaseApproval (address _spender, uint _addedValue) public returns (bool success) {
    allowed[msg.sender][_spender] = allowed[msg.sender][_spender].add(_addedValue);
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  function decreaseApproval (address _spender, uint _subtractedValue) public returns (bool success) {
    uint oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue > oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

}

/**
 * @title Ownable
 * @dev The Ownable contract has an owner address, and provides basic authorization control
 * functions, this simplifies the implementation of "user permissions".
 */
contract Ownable {
  address public owner;


  event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);


  /**
   * @dev The Ownable constructor sets the original `owner` of the contract to the sender
   * account.
   */
  function Ownable() {
    owner = msg.sender;
  }


  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }


  /**
   * @dev Allows the current owner to transfer control of the contract to a newOwner.
   * @param newOwner The address to transfer ownership to.
   */
  function transferOwnership(address newOwner) onlyOwner public {
    require(newOwner != address(0));
    OwnershipTransferred(owner, newOwner);
    owner = newOwner;
  }

}

/**
 * @title Mintable token
 * @dev Simple ERC20 Token example, with mintable token creation
 * @dev Issue: * https://github.com/OpenZeppelin/zeppelin-solidity/issues/120
 * Based on code by TokenMarketNet: https://github.com/TokenMarketNet/ico/blob/master/contracts/MintableToken.sol
 */

contract MintableToken is StandardToken, Ownable {
  event Mint(address indexed to, uint256 amount);
  event MintFinished();

  bool public mintingFinished = false;


  modifier canMint() {
    require(!mintingFinished);
    _;
  }

  /**
   * @dev Function to mint tokens
   * @param _to The address that will receive the minted tokens.
   * @param _amount The amount of tokens to mint.
   * @return A boolean that indicates if the operation was successful.
   */
  function mint(address _to, uint256 _amount) onlyOwner canMint public returns (bool) {
    totalSupply = totalSupply.add(_amount);
    balances[_to] = balances[_to].add(_amount);
    Mint(_to, _amount);
    Transfer(address(0), _to, _amount);
    return true;
  }

  /**
   * @dev Function to stop minting new tokens.
   * @return True if the operation was successful.
   */
  function finishMinting() onlyOwner public returns (bool) {
    mintingFinished = true;
    MintFinished();
    return true;
  }
}

contract SimpleCoinToken is MintableToken {
    
  string public constant name = "AntiqMall";
   
  string public constant symbol = "AMT";
    
  uint32 public constant decimals = 18;
    
}

/**
 * @title Crowdsale
 * @dev Crowdsale is a base contract for managing a token crowdsale.
 * Crowdsales have a start and end timestamps, where investors can make
 * token purchases and the crowdsale will assign them tokens based
 * on a token per ETH rate. Funds collected are forwarded to a wallet
 * as they arrive.
 */
contract Crowdsale is Ownable {
  using SafeMath for uint256;

  // The token being sold
  SimpleCoinToken public token;

  // address where funds are collected
  address public wallet;

  // amount of raised in wei
  uint256 public weiRaised;
  
  uint256 public tokensCount;
  
  uint256 public bountyTokensCount;
  
  enum State {
    early_pre_ico,  
    pre_ico,
    ico_w1,
    ico_w2,
    ico_w3,
    ico_w4,
    ico,
    paused,
    finished
  }
  
  State public currentState;
  
  uint256 constant MIN_WEI_VALUE = 1 * 1e16;
  
  uint256 constant PRE_ICO_SALE_VALUE = 10;
  uint256 constant EARLY_PRE_ICO_SALE_VALUE_1 = 4;
  uint256 constant EARLY_PRE_ICO_SALE_VALUE_2 = 8;
  uint256 constant EARLY_PRE_ICO_SALE_VALUE_3 = 15;
  
  uint256 constant EARLY_PRE_ICO_SALE_BONUS_0 = 50;
  uint256 constant EARLY_PRE_ICO_SALE_BONUS_1 = 65;
  uint256 constant EARLY_PRE_ICO_SALE_BONUS_2 = 75;
  uint256 constant EARLY_PRE_ICO_SALE_BONUS_3 = 85;
  uint256 constant EARLY_PRE_ICO_SALE_BONUS_4 = 100;
  
  uint256 constant PRE_ICO_SALE_BONUS = 100;
  uint256 constant PRE_ICO_BONUS = 50;
  uint256 constant ICO_BONUS_W1 = 30;
  uint256 constant ICO_BONUS_W2 = 20;
  uint256 constant ICO_BONUS_W3 = 10;
  uint256 constant ICO_BONUS_W4 =  0;
  uint256 constant ICO_DEFAULT_BONUS = 0;
  uint256 constant PRICE = 1000;
  uint256 constant ICO_TOKENS_LIMIT = 7.5 * 1e6 * 1e18;
  uint256 constant PRE_ICO_TOKENS_LIMIT = 0.6 * 1e6 * 1e18;
  
  uint256 constant RESERVED_TOKENS_PERCENT = 20;


  /**
   * event for token purchase logging
   * @param purchaser who paid for the tokens
   * @param beneficiary who got the tokens
   * @param value weis paid for purchase
   * @param amount amount of tokens purchased
   */
  event TokenPurchase(address indexed purchaser, address indexed beneficiary, uint256 value, uint256 amount);
  
  modifier saleIsOn() {
    require(currentState <= State.ico);
    _;
  }

  function Crowdsale() {
    token = createTokenContract();
    currentState = State.paused;
    wallet = msg.sender;
    tokensCount = 0;
    bountyTokensCount = 0;
  }

  // creates the token to be sold.
  // override this method to have crowdsale of a specific mintable token.
  function createTokenContract() internal returns (SimpleCoinToken) {
    return new SimpleCoinToken();
  }

  function setIcoState(State _newState) public onlyOwner {
    currentState = _newState;
  }
  
  function mintBountyTokens(address _wallet) public onlyOwner payable {
    uint256 tokens = tokensCount.sub(bountyTokensCount);
    tokens = tokens.mul(25).div(100);
    tokens = tokens.sub(bountyTokensCount);
    
    require(tokens >= 1);
      
    // update state
    tokensCount = tokensCount.add(tokens);
    bountyTokensCount = bountyTokensCount.add(tokens);

    token.mint(_wallet, tokens);
  }

  // fallback function can be used to buy tokens
  function () payable {
    buyTokens(msg.sender);
  }

  // low level token purchase function
  function buyTokens(address beneficiary) public saleIsOn payable {
    require(beneficiary != address(0));
    require(msg.value != 0);
    require(msg.value >= MIN_WEI_VALUE);
    
    uint256 limit = getLimit();
    
    uint256 weiAmount = msg.value;
    
    // calculate token amount to be created
    uint256 tokens = weiAmount.mul(PRICE);
    uint256 bonusTokens = getBonusTokens(tokens, weiAmount);
    
    tokens = tokens.add(bonusTokens);
    
    require(limit >= tokensCount.add(tokens).sub(bountyTokensCount));

    // update state
    weiRaised = weiRaised.add(weiAmount);
    tokensCount = tokensCount.add(tokens);

    token.mint(beneficiary, tokens);
    TokenPurchase(msg.sender, beneficiary, weiAmount, tokens);

    forwardFunds();
  }
  
  function getBonusTokens(uint256 _tokens, uint256 _weiAmount) private returns(uint256 _bonusTokens) {
    uint256 bonusTokens = ICO_DEFAULT_BONUS;
    if(currentState == State.pre_ico) {
      bonusTokens = _tokens.div(100).mul(PRE_ICO_BONUS);
      
      if(_weiAmount >= PRE_ICO_SALE_VALUE.mul(1e18)) {
        bonusTokens = _tokens.div(100).mul(PRE_ICO_SALE_BONUS);
      }
    }
    if(currentState == State.ico_w1) {
      bonusTokens = _tokens.div(100).mul(ICO_BONUS_W1);
    }
    if(currentState == State.ico_w2) {
      bonusTokens = _tokens.div(100).mul(ICO_BONUS_W2);
    }
    if(currentState == State.ico_w3) {
      bonusTokens = _tokens.div(100).mul(ICO_BONUS_W3);
    }
    if(currentState == State.ico_w4) {
      bonusTokens = _tokens.div(100).mul(ICO_BONUS_W4);
    }
    
    if(currentState == State.early_pre_ico) {
        
        bonusTokens = _tokens.div(100).mul(EARLY_PRE_ICO_SALE_BONUS_0);
        
        if(_weiAmount >= 0.5*1e18 && _weiAmount < EARLY_PRE_ICO_SALE_VALUE_1.mul(1e18)) {
            bonusTokens = _tokens.div(100).mul(EARLY_PRE_ICO_SALE_BONUS_1);
        }
        if(_weiAmount >= EARLY_PRE_ICO_SALE_VALUE_1.mul(1e18) && _weiAmount < EARLY_PRE_ICO_SALE_VALUE_2.mul(1e18)) {
            bonusTokens = _tokens.div(100).mul(EARLY_PRE_ICO_SALE_BONUS_2);
        }
        if(_weiAmount >= EARLY_PRE_ICO_SALE_VALUE_2.mul(1e18) && _weiAmount < EARLY_PRE_ICO_SALE_VALUE_3.mul(1e18)) {
            bonusTokens = _tokens.div(100).mul(EARLY_PRE_ICO_SALE_BONUS_3);
        }
        if(_weiAmount >= EARLY_PRE_ICO_SALE_VALUE_3.mul(1e18)) {
            bonusTokens = _tokens.div(100).mul(EARLY_PRE_ICO_SALE_BONUS_4);
        }
    }
    
    return bonusTokens;
  }
  
  function getLimit() private returns(uint256 _limit) {
    if(currentState <= State.pre_ico) {
      return PRE_ICO_TOKENS_LIMIT;
    }
    
    return ICO_TOKENS_LIMIT.sub(ICO_TOKENS_LIMIT.mul(RESERVED_TOKENS_PERCENT).div(100));
  }

  // send ether to the fund collection wallet
  // override to create custom fund forwarding mechanisms
  function forwardFunds() internal {
    wallet.transfer(msg.value);
  }

}